Examples of Modeling Count Outcomes in SAS and STATA

The data for this example come from a recent study of the effects of emotion regulation strategy (none=control, cognitive reappraisal, or suppression) on aggressive behavior in persons with or without a history of intimate partner violence (IPV). The analysis was planned as a 2x3 between-groups ANCOVA with factors for strategy (3) and IPV (2), with neutral condition aggression as a covariate.

SAS Syntax and Output for Data Manipulation:

```
DATA work.example11; SET work.example11;
       LABEL IPV=          "Intimate Partner Violence (0=N,1=Y)"
             Neutral=      "Aggressive during Neutral Condition (0=N,1=Y)"
             ERconds=      "Condition (1=None, 2=CogR, 3=Supp)"
             Aggr=         "Aggressive Verbalizations Outcome"; RUN;
* Create labels for groups;
PROC FORMAT; VALUE Fercond 1="1_None" 2="2_CogR" 3="3_Supp"; RUN;
PROC UNIVARIATE NOPRINT DATA=work.example11; VAR Aggr; HISTOGRAM Aggr; RUN;
```

STATA Syntax and Output for Data Manipulation:

```
* Import data, transform variables, apply value formats
use "$filesave\example11.dta", clear
label define fercond 1 "1_None" 2 "2_CogR" 3 "3_Supp"
label values ercond fercond
display as result "Histogram for Aggressive Verbalizations Outcome"
hist aggr
```



Look how non-aggressive our sample is! That's great for them, but not so good if we expect to use a general linear model (i.e., ANOVA) to analyze this outcome…

However, this is only the marginal distribution of Y. Maybe the residuals will look more normal?



These are the condition means our model is trying to capture….

What we will see in this example is that the means per condition will stay the same across models.

What changes are the inferences about their differences (which comes from the standard error of the difference, which comes from the model for the variance).

Model Predicting Aggressive Verbalizations using Normal Distribution and Identity Link (ANCOVA, usually estimated with least squares, here with ML using MIXED)

$$E(Aggr_i) = \beta_0 + \beta_1 Neutral_i + \beta_2 IPV_i + \beta_3 NoneVsCog_i + \beta_4 NoneVsSupp_i + \beta_5 IPV_i * NoneVsCog_i + \beta_6 IPV_i * NoneVsSupp_i + e_i$$

```
display as result "GLM of Aggressive Verbalizations Outcome"
mixed aggr c.neutral c.ipv##ib(last).ercond , mle
estat ic, n(225),

* Cell means and simple effects of condition per IPV
margins i.ercond, at(c.neutral=0 c.ipv=0) // cell means for IPV=0
margins i.ercond, at(c.neutral=0 c.ipv=0) pwcompare(pveffects) // cell mean diffs
margins i.ercond, at(c.neutral=0 c.ipv=1) // cell means for IPV=1
margins i.ercond, at(c.neutral=0 c.ipv=1) pwcompare(pveffects) // cell mean diffs

* Simple effects of IPV per condition
lincom c.ipv + c.ipv#i1.ercond  // No/Yes IPV Difference: No Instruction
lincom c.ipv + c.ipv#i2.ercond  // No/Yes IPV Difference: Cognitive Reappraisal
lincom c.ipv + c.ipv#i3.ercond  // No/Yes IPV Difference: Suppression

* Specific interaction contrasts
contrast  c.ipv#i.ercond,          // multivariate test of interaction
contrast {c.ipv#i.ercond -1  1  0}, pveffects // No/Yes IPV by None/Cog Interaction
contrast {c.ipv#i.ercond -1  0  1}, pveffects // No/Yes IPV by None/Sup Interaction
contrast {c.ipv#i.ercond  0 -1  1}, pveffects // No/Yes IPV by Cog/Sup Interaction
```

```
TITLE1 "GLM of Aggressive Verbalizations Outcome";
PROC MIXED DATA=work.example11 NOCLPRINT COVTEST NAMELEN=100 METHOD=ML;
CLASS ERcond;
MODEL Aggr = Neutral IPV ERcond IPV*ERcond / CHISQ RESIDUAL SOLUTION;

* Simple effects of condition per IPV;
LSMEANS ERcond / AT (Neutral IPV) = (0 0) DIFF=ALL; * Cell means and diffs for IPV=0;
LSMEANS ERcond / AT (Neutral IPV) = (0 1) DIFF=ALL; * Cell means and diffs for IPV=1;

* Simple effects of IPV per condition;
ESTIMATE "No/Yes IPV Difference: No Instruct"  IPV 1 IPV*ERcond  1  0  0;
ESTIMATE "No/Yes IPV Difference: Cog Reapp"    IPV 1 IPV*ERcond  0  1  0;
ESTIMATE "No/Yes IPV Difference: Suppression"  IPV 1 IPV*ERcond  0  0  1;

* Specific interaction contrasts;
ESTIMATE "No/Yes IPV by None/Cog Interaction"        IPV*ERcond -1  1  0;
ESTIMATE "No/Yes IPV by None/Sup Interaction"        IPV*ERcond -1  0  1;
ESTIMATE "No/Yes IPV by Cog/Sup  Interaction"        IPV*ERcond  0 -1  1;
RUN; TITLE1;
```

SAS Output:

Covariance Parameter Estimates

Cov Parm	Estimate	Standard Error	Z Value	Pr > Z
Residual	11.1803	1.0541	10.61	<.0001

Fit Statistics

-2 Log Likelihood	1181.7
AIC (smaller is better)	1197.7
AICC (smaller is better)	1198.4
BIC (smaller is better)	1225.0

Solution for Fixed Effects

Effect	1=None, 2=CogR, 3=Supp	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept		2.3272	0.4553	218	5.11	<.0001
Neutral		1.3297	0.8306	218	1.60	0.1109
IPV		1.9649	0.8251	218	2.38	0.0181
ERcond	1_None	0.2643	0.6496	218	0.41	0.6845
ERcond	2_CogR	-1.5031	0.6483	218	-2.32	0.0213
ERcond	3_Supp	0	.	.	.	.
IPV*ERcond	1_None	-1.6988	1.1910	218	-1.43	0.1552
IPV*ERcond	2_CogR	-2.7720	1.2031	218	-2.30	0.0222
IPV*ERcond	3_Supp	0	.	.	.	.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
Neutral	1	218	2.56	2.56	0.1094	0.1109
IPV	1	218	0.92	0.92	0.3380	0.3391
ERcond	2	218	8.63	4.31	0.0134	0.0145
IPV*ERcond	2	218	5.46	2.73	0.0653	0.0676

p = .0676, seriously?

Least Squares Means

Effect	1=None, 2=CogR, 3=Supp	Neutral	IPV	Estimate	Standard Error	DF	t Value	Pr > |t|
ERcond	1_None	0.00	0.00	2.5915	0.4659	218	5.56	<.0001
ERcond	2_CogR	0.00	0.00	0.8242	0.4636	218	1.78	0.0768
ERcond	3_Supp	0.00	0.00	2.3272	0.4553	218	5.11	<.0001
ERcond	1_None	0.00	1.00	2.8577	0.7392	218	3.87	0.0001
ERcond	2_CogR	0.00	1.00	0.01703	0.7523	218	0.02	0.9820
ERcond	3_Supp	0.00	1.00	4.2921	0.6904	218	6.22	<.0001

Differences of Least Squares Means

Effect	1=None, 2=CogR, 3=Supp	1=None, 2=CogR, 3=Supp	Neutral	IPV	Estimate	Standard Error	DF	t Value	Pr > |t|
ERcond	1_None	2_CogR	0.00	0.00	1.7674	0.6497	218	2.72	0.0071
ERcond	1_None	3_Supp	0.00	0.00	0.2643	0.6496	218	0.41	0.6845
ERcond	2_CogR	3_Supp	0.00	0.00	-1.5031	0.6483	218	-2.32	0.0213
ERcond	1_None	2_CogR	0.00	1.00	2.8406	1.0453	218	2.72	0.0071
ERcond	1_None	3_Supp	0.00	1.00	-1.4345	0.9992	218	-1.44	0.1526
ERcond	2_CogR	3_Supp	0.00	1.00	-4.2751	1.0126	218	-4.22	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
No/Yes IPV Difference: No Instruct	0.2661	0.8631	218	0.31	0.7581
No/Yes IPV Difference: Cog Reapp	-0.8071	0.8777	218	-0.92	0.3588
No/Yes IPV Difference: Suppression	1.9649	0.8251	218	2.38	0.0181
No/Yes IPV by None/Cog Interaction	-1.0733	1.2303	218	-0.87	0.3840
No/Yes IPV by None/Sup Interaction	1.6988	1.1910	218	1.43	0.1552
No/Yes IPV by Cog/Sup Interaction	2.7720	1.2031	218	2.30	0.0222

What about that whole non-normal residuals thing? Yep, still an issue… what to do instead?



Model Predicting Aggressive Verbalizations using Poisson Distribution and Log Link

$$Log[E(Aggr_i)] = \beta_0 + \beta_1 Neutral_i + \beta_2 IPV_i + \beta_3 NoneVsCog_i + \beta_4 NoneVsSupp_i$$
$$+ \beta_5 IPV_i * NoneVsCog_i + \beta_6 IPV_i * NoneVsSupp_i$$

```
display as result "Poisson Model of Aggressive Verbalizations Outcome"
mepoisson aggr c.neutral c.ipv##ib(last).ercond ,
estat ic, n(225),

* Cell means in log count (model version) and simple effects of condition per IPV
margins i.ercond, at(c.neutral=0 c.ipv=0) predict(xb) // cell means for IPV=0
margins i.ercond, at(c.neutral=0 c.ipv=0) predict(xb) pwcompare(pveffects) // mean diffs
margins i.ercond, at(c.neutral=0 c.ipv=1) predict(xb) // cell means for IPV=1
margins i.ercond, at(c.neutral=0 c.ipv=1) predict(xb) pwcompare(pveffects) // mean diffs

* Cell means in predicted count
margins i.ercond, at(c.neutral=0 c.ipv=0) // cell means for IPV=0
margins i.ercond, at(c.neutral=0 c.ipv=1) // cell means for IPV=1

* Simple effects of IPV per condition (in log count)
lincom c.ipv + c.ipv#i1.ercond  // No/Yes IPV Difference: No Instruction
lincom c.ipv + c.ipv#i2.ercond  // No/Yes IPV Difference: Cognitive Reappraisal
lincom c.ipv + c.ipv#i3.ercond  // No/Yes IPV Difference: Suppression

* Specific interaction contrasts (in log count)
contrast  c.ipv#i.ercond,          // multivariate test of interaction
contrast {c.ipv#i.ercond -1  1  0}, pveffects // No/Yes IPV by None/Cog Interaction
contrast {c.ipv#i.ercond -1  0  1}, pveffects // No/Yes IPV by None/Sup Interaction
contrast {c.ipv#i.ercond  0 -1  1}, pveffects // No/Yes IPV by Cog/Sup Interaction
```

```
TITLE1 "Poisson Model of Aggressive Verbalizations";
PROC GLIMMIX DATA=work.example11 NOCLPRINT NAMELEN=100;
CLASS ERcond;
MODEL Aggr = Neutral IPV ERcond IPV*ERcond / CHISQ SOLUTION LINK=LOG DIST=POISSON;
* Cell means and simple effects of condition per IPV;
LSMEANS ERcond / ILINK AT (Neutral IPV) = (0 0) DIFF=ALL; * Cell means and diffs for IPV=0;
LSMEANS ERcond / ILINK AT (Neutral IPV) = (0 1) DIFF=ALL; * Cell means and diffs for IPV=1;
* Simple effects of IPV per condition;
ESTIMATE "No/Yes IPV Difference: No Instruct"   IPV 1 IPV*ERcond  1  0  0;
ESTIMATE "No/Yes IPV Difference: Cog Reapp"     IPV 1 IPV*ERcond  0  1  0;
ESTIMATE "No/Yes IPV Difference: Suppression"   IPV 1 IPV*ERcond  0  0  1;
* Specific interaction contrasts;
ESTIMATE "No/Yes IPV by None/Cog Interaction"         IPV*ERcond -1  1  0;
ESTIMATE "No/Yes IPV by None/Sup Interaction"         IPV*ERcond -1  0  1;
ESTIMATE "No/Yes IPV by Cog/Sup  Interaction"         IPV*ERcond  0 -1  1;
RUN; TITLE1;
```

SAS Output:

Fit Statistics

-2 Log Likelihood	1144.96
AIC (smaller is better)	1158.96
AICC (smaller is better)	1159.48
BIC (smaller is better)	1182.87
CAIC (smaller is better)	1189.87
HQIC (smaller is better)	1168.61
Pearson Chi-Square	1016.55
Pearson Chi-Square / DF	4.66

Parameter Estimates

Effect	1=None, 2=CogR, 3=Supp	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept		0.8440	0.08883	218	9.50	<.0001
Neutral		0.4740	0.1366	218	3.47	0.0006
IPV		0.5777	0.1330	218	4.34	<.0001
ERcond	1_None	0.09986	0.1227	218	0.81	0.4167
ERcond	2_CogR	-0.9673	0.1686	218	-5.74	<.0001
ERcond	3_Supp	0	.	.	.	.
IPV*ERcond	1_None	-0.4903	0.1998	218	-2.45	0.0149
IPV*ERcond	2_CogR	-2.4104	0.6093	218	-3.96	0.0001
IPV*ERcond	3_Supp	0	.	.	.	.

Type III Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
Neutral	1	218	12.05	12.05	0.0005	0.0006
IPV	1	218	3.45	3.45	0.0631	0.0645
ERcond	2	218	43.87	21.94	<.0001	<.0001
IPV*ERcond	2	218	19.29	9.64	<.0001	<.0001

Now that's more like it! ☺

ERcond Least Squares Means

1=None, 2=CogR, 3=Supp	Neutral	IPV	Estimate	Standard Error	DF	t Value	Pr > |t|	Mean	Standard Error Mean
1_None	0.00	0.00	0.9439	0.08560	218	11.03	<.0001	2.5700	0.2200
2_CogR	0.00	0.00	-0.1232	0.1437	218	-0.86	0.3922	0.8841	0.1271
3_Supp	0.00	0.00	0.8440	0.08883	218	9.50	<.0001	2.3257	0.2066

```
1_None        0.00     1.00      1.0313     0.1283     218      8.04      <.0001      2.8047      0.3598
2_CogR        0.00     1.00     -1.9560     0.5777     218     -3.39      0.0008      0.1414     0.08170
3_Supp        0.00     1.00      1.4217     0.09998    218     14.22      <.0001      4.1442      0.4143


                            Differences of ERcond Least Squares Means
1=None,       1=None,
2=CogR,       2=CogR,                                        Standard
3=Supp        3=Supp        Neutral       IPV    Estimate       Error       DF    t Value    Pr > |t|

1_None        2_CogR           0.00      0.00      1.0671      0.1654      218       6.45      <.0001
1_None        3_Supp           0.00      0.00     0.09986      0.1227      218       0.81      0.4167
2_CogR        3_Supp           0.00      0.00     -0.9673      0.1686      218      -5.74      <.0001

1_None        2_CogR           0.00      1.00      2.9873      0.5908      218       5.06      <.0001
1_None        3_Supp           0.00      1.00     -0.3904      0.1581      218      -2.47      0.0143
2_CogR        3_Supp           0.00      1.00     -3.3777      0.5854      218      -5.77      <.0001

                                        Estimates
                                                   Standard
Label                                  Estimate       Error       DF    t Value    Pr > |t|

No/Yes IPV Difference: No Instruct      0.08739      0.1505      218       0.58      0.5621
No/Yes IPV Difference: Cog Reapp        -1.8328      0.5948      218      -3.08      0.0023
No/Yes IPV Difference: Suppression       0.5777      0.1330      218       4.34      <.0001

No/Yes IPV by None/Cog Interaction      -1.9202      0.6135      218      -3.13      0.0020
No/Yes IPV by None/Sup Interaction       0.4903      0.1998      218       2.45      0.0149
No/Yes IPV by Cog/Sup  Interaction       2.4104      0.6093      218       3.96      0.0001
```

Model Predicting Aggressive Verbalizations using Negative Binomial Distribution and Log Link

$$Log[E(Aggr_i)] = \beta_0 + \beta_1 Neutral_i + \beta_2 IPV_i + \beta_3 NoneVsCog_i + \beta_4 NoneVsSupp_i$$
$$+ \beta_5 IPV_i * NoneVsCog_i + \beta_6 IPV_i * NoneVsSupp_i + e_i$$

```
display as result "Negative Binomial Model of Aggressive Verbalizations Outcome"
menbreg aggr c.neutral c.ipv##ib(last).ercond ,
estat ic, n(225),

* Cell means in log count (model version) and simple effects of condition per IPV
margins i.ercond, at(c.neutral=0 c.ipv=0) predict(xb) // cell means for IPV=0
margins i.ercond, at(c.neutral=0 c.ipv=0) predict(xb) pwcompare(pveffects) // mean diffs
margins i.ercond, at(c.neutral=0 c.ipv=1) predict(xb) // cell means for IPV=1
margins i.ercond, at(c.neutral=0 c.ipv=1) predict(xb) pwcompare(pveffects) // mean diffs

* Cell means in predicted count
margins i.ercond, at(c.neutral=0 c.ipv=0) // cell means for IPV=0
margins i.ercond, at(c.neutral=0 c.ipv=1) // cell means for IPV=1

* Simple effects of IPV per condition (in log count)
lincom c.ipv + c.ipv#i1.ercond  // No/Yes IPV Difference: No Instruction
lincom c.ipv + c.ipv#i2.ercond  // No/Yes IPV Difference: Cognitive Reappraisal
lincom c.ipv + c.ipv#i3.ercond  // No/Yes IPV Difference: Suppression

* Specific interaction contrasts (in log count)
contrast  c.ipv#i.ercond,          // multivariate test of interaction
contrast {c.ipv#i.ercond -1  1  0}, pveffects // No/Yes IPV by None/Cog Interaction
contrast {c.ipv#i.ercond -1  0  1}, pveffects // No/Yes IPV by None/Sup Interaction
contrast {c.ipv#i.ercond  0 -1  1}, pveffects // No/Yes IPV by Cog/Sup Interaction
```

```
TITLE1 "Negative Binomial Model of Aggressive Verbalizations";
PROC GLIMMIX DATA=work.example11 NOCLPRINT NAMELEN=100;
CLASS ERcond;
MODEL Aggr = Neutral IPV ERcond IPV*ERcond / CHISQ SOLUTION LINK=LOG DIST=NEGBIN;

* Cell means and simple effects of condition per IPV;
LSMEANS ERcond / ILINK AT (Neutral IPV) = (0 0) DIFF=ALL; * Cell means and diffs for IPV=0;
LSMEANS ERcond / ILINK AT (Neutral IPV) = (0 1) DIFF=ALL; * Cell means and diffs for IPV=1;

* Simple effects of IPV per condition;
ESTIMATE "No/Yes IPV Difference: No Instruct"   IPV 1 IPV*ERcond  1  0  0;
ESTIMATE "No/Yes IPV Difference: Cog Reapp"     IPV 1 IPV*ERcond  0  1  0;
ESTIMATE "No/Yes IPV Difference: Suppression"   IPV 1 IPV*ERcond  0  0  1;

* Specific interaction contrasts;
ESTIMATE "No/Yes IPV by None/Cog Interaction"         IPV*ERcond -1  1  0;
ESTIMATE "No/Yes IPV by None/Sup Interaction"         IPV*ERcond -1  0  1;
ESTIMATE "No/Yes IPV by Cog/Sup  Interaction"         IPV*ERcond  0 -1  1;
RUN; TITLE1;
```

SAS Output:

Fit Statistics

-2 Log Likelihood	817.95
AIC (smaller is better)	833.95
AICC (smaller is better)	834.61
BIC (smaller is better)	861.28
CAIC (smaller is better)	869.28
HQIC (smaller is better)	844.98
Pearson Chi-Square	247.87
Pearson Chi-Square / DF	1.14

Poisson model $-2LL = 1144.96$

$-2\Delta LL(df = 1) = 1146.96 - 817.95 = 329.01$, $p = 1.57911E\text{-}73$

So the model fits significantly better from adding a "dispersion" (scale) parameter that allows the variance to exceed the mean.

Parameter Estimates

Effect	1=None, 2=CogR, 3=Supp	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept		0.8512	0.1927	218	4.42	<.0001
Neutral		0.4493	0.3634	218	1.24	0.2177
IPV		0.5671	0.3391	218	1.67	0.0959
ERcond	1_None	0.07835	0.2760	218	0.28	0.7768
ERcond	2_CogR	-0.9583	0.2963	218	-3.23	0.0014
ERcond	3_Supp	0	.	.	.	.
IPV*ERcond	1_None	-0.4379	0.4941	218	-0.89	0.3765
IPV*ERcond	2_CogR	-2.4299	0.7596	218	-3.20	0.0016
IPV*ERcond	3_Supp	0	.	.	.	.
Scale		**1.5789**	**0.2408**	.	.	.

In STATA the scale factor is given as log(scale) instead, but it's the same idea of "dispersion" needed.

Type III Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
Neutral	1	218	1.53	1.53	0.2164	0.2177
IPV	1	218	1.91	1.91	0.1668	0.1682
ERcond	2	218	14.53	7.26	0.0007	0.0009
IPV*ERcond	2	218	10.24	5.12	0.0060	0.0067

ERconds Least Squares Means

1=None, 2=CogR, 3=Supp	Neutral	IPV	Estimate	Standard Error	DF	t Value	Pr > |t|	Mean	Standard Error Mean
1_None	0.00	0.00	0.9296	0.1983	218	4.69	<.0001	2.5334	0.5024
2_CogR	0.00	0.00	-0.1071	0.2254	218	-0.47	0.6353	0.8985	0.2025
3_Supp	0.00	0.00	0.8512	0.1927	218	4.42	<.0001	2.3425	0.4514
1_None	0.00	1.00	1.0588	0.3043	218	3.48	0.0006	2.8829	0.8772
2_CogR	0.00	1.00	-1.9699	0.6469	218	-3.05	0.0026	0.1395	0.09022
3_Supp	0.00	1.00	1.4183	0.2796	218	5.07	<.0001	4.1301	1.1547

Differences of ERcond Least Squares Means

(1=None, 2=CogR, 3=Supp)	(1=None, 2=CogR, 3=Supp)	Neutral	IPV	Estimate	Standard Error	DF	t Value	Pr > |t|
1_None	2_CogR	0.00	0.00	1.0366	0.2966	218	3.49	0.0006
1_None	3_Supp	0.00	0.00	0.07835	0.2760	218	0.28	0.7768
2_CogR	3_Supp	0.00	0.00	-0.9583	0.2963	218	-3.23	0.0014
1_None	2_CogR	0.00	1.00	3.0286	0.7113	218	4.26	<.0001
1_None	3_Supp	0.00	1.00	-0.3595	0.4082	218	-0.88	0.3794
2_CogR	3_Supp	0.00	1.00	-3.3882	0.6998	218	-4.84	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
No/Yes IPV Difference: No Instruction	0.1292	0.3578	218	0.36	0.7184
No/Yes IPV Difference: Cognitive Reappraisal	-1.8628	0.6830	218	-2.73	0.0069
No/Yes IPV Difference: Suppression	0.5671	0.3391	218	1.67	0.0959
No/Yes IPV by None/Cog Interaction	-1.9920	0.7716	218	-2.58	0.0105
No/Yes IPV by None/Sup Interaction	0.4379	0.4941	218	0.89	0.3765
No/Yes IPV by Cog/Sup Interaction	2.4299	0.7596	218	3.20	0.0016

Also examined Zero-Inflated Poisson and Zero-Inflated Negative Binomial, just to be sure:

```
display as result "Zero-Inflated Poisson Model of Aggressive Verbalizations Outcome"
display as result "Have to switch to single-level (univariate) model"
zip aggr c.neutral c.ipv##ib(last).ercond , inflate(_cons),
estat ic, n(225)

TITLE1 "Zero-Inflated Poisson Model of Aggressive Verbalizations";
PROC GENMOD DATA=work.example11 NAMELEN=100;
CLASS ERcond;
MODEL Aggr = Neutral IPV ERcond IPV*ERcond / LINK=LOG DIST=ZIP;
ZEROMODEL / LINK=LOGIT; RUN; TITLE1;
```

Criteria For Assessing Goodness Of Fit

Criterion	Value
Full Log Likelihood	-489.0974
AIC (smaller is better)	994.1948
AICC (smaller is better)	994.8615
BIC (smaller is better)	1021.5236

−2*489.0974 = −978.19
Previous best model Negative Binomial:
−2LL = 817.95, AIC =833.95, BIC = 861.28

AIC and BIC are higher for ZIP, so NB is still better.

Analysis Of Maximum Likelihood Zero Inflation Parameter Estimates

Parameter	DF	Estimate	Standard Error	Wald 95% Confidence Limits		Wald Chi-Square	Pr > ChiSq
Intercept	1	-0.5461	0.1650	-0.8696	-0.2227	10.95	0.0009

```
*** STATA BLEW UP!!
*display as result "Zero-Inflated Negative Binomial of Aggressive Verbalizations Outcome"
*display as result "Have to switch to single-level (univariate) model"
*zinb aggr c.neutral c.ipv##ib(last).ercond , inflate(_cons),
*estat ic, n(225)

TITLE1 "Zero-Inflated Negative Binomial Model of Aggressive Verbalizations";
PROC GENMOD DATA=work.example11 NAMELEN=100;
CLASS ERcond;
MODEL Aggr = Neutral IPV ERcond IPV*ERcond / LINK=LOG DIST=ZINB;
ZEROMODEL / LINK=LOGIT; RUN; TITLE1;
```

Criteria For Assessing Goodness Of Fit

Criterion	Value
Full Log Likelihood	-408.9731
AIC (smaller is better)	835.9463
AICC (smaller is better)	836.7835
BIC (smaller is better)	866.6912

−2*108.9731 = −817.95
Previous best model Negative Binomial:
−2LL = 817.95, AIC =833.95, BIC = 861.28

AIC and BIC are higher for ZINB, so NB is enough.

Analysis Of Maximum Likelihood Zero Inflation Parameter Estimates

Parameter	DF	Estimate	Standard Error	Wald 95% Confidence Limits		Wald Chi-Square	Pr > ChiSq
Intercept	1	-22.2405	39404.29	-77253.2	77208.75	0.00	0.9995

The logit of being an "extra 0" = −22 (with a standard error of 40,000)!
This translates into a probability of p =.0000000002789 of being an "extra 0".
In other words, there are no extra 0's in this distribution not already predicted by the negative binomial.